

Expert의 레이아웃 작업 자동화를 위한 Lisa 스크립트 작성

소개

Expert에서 레이아웃 설계 작업의 많은 부분을 Lisa 명령어를 사용하여 자동화할 수 있습니다. 반복적인 수동 작업에 대해 Lisa Expert Interface (XI) 스크립트를 만들어, 수작업에 의한 레이아웃의 오류 발생 가능성을 줄여서 프로젝트 시간을 단축하고 레이아웃 품질을 높일 수 있습니다. Expert 메뉴의 명령어를 Lisa 명령으로 복제할 수 있으므로, XI 스크립트를 작성하여 거의 모든 레이아웃 작업을 자동화할 수 있습니다. 일단 작성된 스크립트는 상이한 기술 프로세스 노드에 대한 스크립트를 포함하므로, 다양한 설계 프로젝트에서 사용할 수 있도록 신속하게 수정할 수 있습니다.

두 개의 XI 스크립트 예제를 제공하여, 시간이 오래 걸리는 레이아웃 수작업을 Lisa 스크립트로 처리하는 방법을 설명합니다. 예에서 설계 프로젝트의 레이아웃 단계에서 다양한 작업에 적용할 수 있는 몇 가지 기본 Lisa 명령을 포함합니다.

Lisa XI 스크립트 생성

Expert는 메뉴에서 열 수 있는 XI 스크립트 에디터를 내장하고 있습니다. 그림 1의 Tools >> XI Script >> Script Panel입니다. Expert 사용자 설명서는 XI 스크립트에 대한 설명과 Lisa 명령을 사용하여 Expert 기능에 액세스할 수 있는 구문을 포함합니다. 스크립트가 작성되면, 스크립트 패널의 도구 모음에서 "Run" 아이콘을 클릭하거나 메뉴에서 Script >> Run을 선택하여 실행할 수 있습니다.




그림 1. XI 스크립트 패널 실행

메모리 어레이의 텍스트 라벨 예제 스크립트

그림 2의 첫 번째 XI 스크립트 예는 메모리 셀 어레이를 포함하는 사용자 지정 레이아웃 셀을 열어서, 메모리 비트 라인 및 워드 라인 신호 경로에 일정하게 증가하는 텍스트 라벨을 추가하는 짧은 Lisa 스크립트입니다. 신호 노드에 텍스트 라벨을 사용하면 LVS 디버깅을 크게 개선할 수 있지만, 메모리 설계에서 대규모 어레이에 이러한 라벨을 수동으로 배치하려면 많은 시간이 소요될 수 있습니다.

스크립트의 시작 부분에서 변수를 정의합니다. 변수 정의 구문 다음에 "cell open" 명령어를 사용하여 편집할 메모리 어레이 레이아웃을 엽니다. 이 스크립트는 프로젝트 데이터베이스 .eld 파일을 로드하지 않으므로, 스크립트를 실행하기 전에 데이터베이스를 로드해야 합니다. .eld 파일은 다음 명령 구문을 사용하여 XI 스크립트에서 로드합니다:

```
PROJECT LOAD proj_name [tech_name] [/outputname=output_name] [/readonly]
```

```
! This script labels row and column bit line
! and word line with text labels
bl_pitch= 7.4;
wl_pitch= 7.8;
Xorg=-1.2;
Yorg=0;
bl_num=0;
wl_num=0;
cols=64;
rows=64;
cell_name="array_top";
cell open (cell_name);
LOOP BEGIN
  !bit line loop
  IF (bl_num EQL cols) THEN (leave loop);
  text (Xorg+bl_num*bl_pitch) (Yorg-3) ("bl_"&bl_num)
  /height = 2 /width = 0.8 /bottomcenter /layer = "metal1";
  bl_num=bl_num+1;
END;
  !end of bit line loop
LOOP BEGIN
  !word line loop
  IF (wl_num EQL cols) THEN (leave loop);
  text (Xorg) (Yorg+wl_num*wl_pitch) ("wl_"&wl_num)
  /height = 2 /width = 0.8 /bottomright /layer = "poly";
  wl_num=wl_num+1;
END;
  !end of word line loop
```

그림 2. 텍스트 라벨 XI 스크립트




그림 3. 라벨 어레이의 레이아웃 뷰

다음은 각 비트 라인 열을 반복하여 메탈1 수직 비트 라인에 텍스트 라벨을 추가하는 "LOOP BEGIN" 문입니다. 텍스트는 "bl_0"으로 시작되며 반복을 통해 매번 증가합니다. 동일한 절차를 통해 다중 워드 라인 행에 텍스트 라벨을 배치합니다. 결과 레이아웃의 확대 보기를 그림 3에서 보여줍니다.

컨택 검색 및 예제 스크립트 교체

두 번째 예제 XI 스크립트는 특정 크기의 컨택 모양을 검색하고, 해당 도형을 새 컨택을 포함하는 레이아웃 셀의 인스턴스로 교체하여 레이아웃 데이터베이스에서 컨택 형상의 크기를 조정합니다. 그림 4에서 이 예제의 코드를 보여줍니다.

현재 로드된 레이아웃 데이터베이스의 모든 셀 목록은 "get_library_cell_list" 명령을 사용하여 배열 변수 "cells"에 저장합니다. 그런 다음 "cells" 배열 변수의 각 항목을 반복하고 특정 검색 기준으로 "find objects" 명령을 사용하여 0.6um 크기로 그려진 모든 컨택에 대해 검색하는 루프가 만들어집니다. "find objects" 검색 결과가 있을 경우, 즉 셀에서 0.6x0.6um 컨택을 찾을 경우 두 번째 중첩 루프가 시작됩니다. 루프 내에서 "Select Object" 명령은 발견한 각각의 형상을 선택하여 삭제합니다. 검색에서 반환된 X 및 Y 위치 값을 사용하여 정확하게 0.4um 크기의 컨택 인스턴스를 삭제된 도형의 원점에 배치합니다.

```
! Script to replace a flat contact shape with
! an instance of a smaller contact

cells = get_library_cell_list(""); ! store all cells in array "cells"
i=1;
LOOP BEGIN !loop through all cells
  cell open (cells[i]);
  DISPLAY(cells[i]);
  cont_Op6 = (find objects (SEARCH_BOX)
    /criteria = ({search_criterion_create(OA_LAYER, "contact", EQ),
      search_criterion_create(OA_SIZEX, 0.6, EQ),
      search_criterion_create(OA_SIZEY, 0.6, EQ)}) /seq_output
  );
  num_cont=cont_Op6.size;
  DISPLAY (num_cont);
  IF ((cont_Op6.size) NEQ 0) THEN BEGIN
    j=1;
    LOOP BEGIN
      Select Object (cont_Op6[j]);
      pt=cont_Op6[j].position;
      DELETE SELECTION;
      instance (pt.x) (pt.y) /cell="cont_Op4";
      j = j+1;
      IF (j EQL (num_cont+1)) THEN (leave loop);
    END;
    !end of loop to delete 0.6um contact and replace
    !with "cont_Op4" instance
    !end of condition where num_cont > 0 in a cell
  END;
  cell close;
  i = i + 1;
  IF (i EQL (cells.size)) THEN (leave loop);
END;
```

그림 4. 컨택을 대체하는 XI 스크립트

그림 4에 사용된 "finding objects" 명령의 예는 도형, 와이어, 텍스트 및 일련의 사용자 정의 검색 기준에 맞는 인스턴스를 찾기 위해 대형 레이아웃 셀 또는 전체 레이아웃 데이터베이스를 검색해야 하는 작업에 적용하도록 수정할 수 있습니다.

결론

Lisa 코드로 XI 스크립트를 생성하여, Expert에서 체계적이고 반복적인 레이아웃 기능을 수행할 수 있습니다. 위에서 언급한 두 가지 예시는 몇 가지 기본적인 Lisa 명령어를 사용하여, 수작업으로 할 경우 시간이 오래 걸리는 설계 작업을 자동화하는 방법을 소개합니다. 프로젝트 전체에서 재사용할 수 있고 미래의 프로젝트에 활용할 수 있도록 XI 스크립트를 만든다면, 설계 레이아웃 단계의 사이클을 단축하는 데 상당한 영향을 미칠 수 있습니다.